In[2]:=
```
SetDirectory["~/KappaLib/"];
<< kappaLib-1.2.m
<< helper.m
```

```
Loading KappaLib v1.2

Loading helper.m..
```

In this notebook we define a 1-parameter family of tensors such that:

1) each kappa is algebraically decomposable.

2) beta^2 -alpha gamma = -1.

3) the Fresnel polynomial is product of a two irreducible quadratic forms.

4) The nonlinear equation for D has no solution

- **Define medium**

In[5]:=
$$\text{kappa} = \text{emMatrixToKappa}\left[\begin{pmatrix} 0 & 0 & 0 & 1 & 0 & -1 \\ 0 & 0 & 0 & -1-3\,\text{b13} & -1 & 1 \\ 1 & 0 & -2 & 6-3\,\text{b13} & -1 & 0 \\ 0 & 1+\text{b13} & 0 & 1 & 0 & -1 \\ 0 & 1 & 0 & 1 & 0 & 0 \\ 0 & 0 & 0 & 1 & 0 & 1 \end{pmatrix}\right];$$

```
(* kappa is not invertible *)
FullSimplify[emDet[kappa]]

(* kappa has no axion component *)
Simplify[emTrace[kappa]]

(* Since this does not simplify to {0}, kappa has a skewon component *)
Union[Flatten[kappa - emPoincare[kappa]]]

(* Since this does not simplify to {0}, kappa has a principal component *)
Union[Flatten[kappa + emPoincare[kappa]]]
```

Out[6]= 0

Out[7]= 0

Out[8]=
$$\left\{-3, -2, -1, 0, 1, 2, 3, \frac{1}{4}(-4-12\,\text{b13}), -1-3\,\text{b13}, 7-3\,\text{b13}, 1+\text{b13}, \frac{1}{4}(4+4\,\text{b13}), -1+\frac{1}{4}(-24+12\,\text{b13})\right\}$$

Out[9]=
$$\left\{-2, -1, 0, 1, 2, -1+\frac{1}{4}(24-12\,\text{b13}), \frac{1}{4}(-4-4\,\text{b13}), -1-3\,\text{b13}, 5-3\,\text{b13}, 1+\text{b13}, \frac{1}{4}(4+12\,\text{b13})\right\}$$

- **Define constants alpha, beta, gamma and bivectors A and B**

In[10]:=
```
alpha = 1;
beta = 0;
gamma = 1;
rho = 1;
```

$$\text{Abivector} = \begin{pmatrix} 0 & 0 & 0 & 0 \\ 0 & 0 & -\frac{1}{2} & 0 \\ 0 & \frac{1}{2} & 0 & -1 \\ 0 & 0 & 1 & 0 \end{pmatrix};$$

$$\text{Bbivector} = \begin{pmatrix} 0 & -1 & b13 & 1 \\ 1 & 0 & -1 & -1 \\ -b13 & 1 & 0 & 3\,(-1+b13) \\ -1 & 1 & -3\,(-1+b13) & 0 \end{pmatrix};$$

- **Note these satisfy beta^2-alpha gamma = 0**

In[16]:=
```
beta ^ 2 - alpha gamma
```

Out[16]= -1

- **Verify that each kappa is algebraically decomposable**

In[17]:=
```
LHS = alpha emIdentityKappa[] +
   beta (kappa + emPoincare[kappa]) + gamma emCompose[emPoincare[kappa], kappa];
RHS = emBiProduct[rho, Abivector, Bbivector] + emBiProduct[rho, Bbivector, Abivector];
Union[Simplify[Flatten[(LHS - RHS)]]]
```

Out[19]= {0}

- **Fresnel polynomial is product of two quadratic forms of signatures (+ + - -)**

In[20]:=
```
coords = {xi0, xi1, xi2, xi3};
fresnel = FullSimplify[emKappaToFresnel[kappa, coords]];
```

In[22]:=

$$\text{gPlus} = \begin{pmatrix} 0 & 0 & 1 & 0 \\ 0 & 2 & -3\,(1+b13) & -2 \\ 1 & -3\,(1+b13) & -6 & 0 \\ 0 & -2 & 0 & 0 \end{pmatrix};$$

$$\text{gMinus} = \begin{pmatrix} 0 & -1-b13 & -2+b13 & 0 \\ -1-b13 & 2 & 2-b13 & 0 \\ -2+b13 & 2-b13 & -2\,(-2+b13) & 1+b13 \\ 0 & 0 & 1+b13 & 2 \end{pmatrix};$$

```
frExp = -1 / 4 (coords.gPlus.coords) (coords.gMinus.coords);
FullSimplify[fresnel - frExp]
```

Out[25]= 0

- **The quadratic forms can have sigantures: (- - + +) and (- - + +)**

In[26]:=
```
point = -15;
Sort[Eigenvalues[gPlus /. b13 → point] // N]
Sort[Eigenvalues[gMinus /. b13 → point] // N]
```

Out[27]= {-44.2434, -0.0539647, 0.0416174, 40.2557}

Out[28]= {-24.3721, -0.442794, 15.0222, 47.7927}

- **The quadratic forms can have sigantures: (- - + +) and (- + + +)**

In[29]:=
```
point = -10;
Sort[Eigenvalues[gPlus /. b13 → point] // N]
Sort[Eigenvalues[gMinus /. b13 → point] // N]
```

Out[30]= {-29.3724, -0.0894153, 0.0599576, 25.4018}

Out[31]= {-16.1176, 0.569584, 10.0333, 33.5147}

- **Show that the quadratic factors are irreducible in the complex polynomials**

In[32]:=
```
v = {v0, v1, v2, v3};
w = {w0, w1, w2, w3};
delta = gPlus - (coords.v) (coords.w);
eqs = Union[Flatten[CoefficientList[delta, coords]]];
GroebnerBasis[eqs, Variables[eqs]]

delta = gMinus - (coords.v) (coords.w);
eqs = Union[Flatten[CoefficientList[delta, coords]]];
GroebnerBasis[eqs, Variables[eqs]]
```

Out[36]= {1}

Out[39]= {1}

Show that equation

D (kappa + beta Id) = 1/2 trace (rho bar(D) otimes D) A + B

has no (complex) solution for D

In[40]:=
```
(*
 * If D=1/2 D^ij d/dx^i /\ d/dx^j is a bivector we represent
 * the coefficients by the anti-symmetric matrix with coefficients
 * of (D^ij)_ij. If kappa is an antisymmetric (2,2)-tensor,
 * then this routine returns coefficients of bivector D (kappa).
 *)
contract[biv_,kappa_] := Table[
 1/2 Sum[biv[[i]][[j]]emReadNormal[kappa,a,b,i,j]
      ,
      {i, 1, 4},{j, 1, 4}
    ]
 ,
 {a, 1, 4}, {b, 1, 4}
]
```

In[41]:=
```
Dbivector = emMatrix["d", 4, Structure → "AntiSymmetric"];
dLHS = contract[Dbivector, kappa + beta emIdentityKappa[]];
dRHS = 1 / 2 emTrace[emBiProduct[rho, Dbivector, Dbivector]] Abivector + Bbivector;
deqs = Union[Flatten[Simplify[dLHS - dRHS]]];
GroebnerBasis[deqs, Variables[deqs]]
```

Out[45]= {1}